
//***************************************************************
//* MACRO: FLIPENV.WCM
//* PURPOSE: Create an envelope  rotated180°.  Some printers have a large
//* trailing margin.  This macro will allows the return address to print at .25"
//***************************************************************
Application (WordPerfect; "WordPerfect"; Default!; "US")
Error(Off!;)
FileNew ()
ReturnAdd = PersonalInfoFormatAddr (AddressBookFormatGetCurrent (Item:Format!))
DialogShow (Dialog:"Address"; Parent:"WordPerfect"; Callback:AddBook)
CallbackWait
DialogDestroy (Dialog:"Address")
PaperSizeDlg ()

// Set margins
MarginTop (MarginWidth: 0.25")
MarginBottom (MarginWidth: 0.25")
MarginLeft (MarginWidth: 1")
MarginRight (MarginWidth: 0.25")

// Create TextBox for Return address
BoxCreate (TextBox!)
BoxBorder (Style: 0)
BoxAttachTo (Anchor:Page!)
BoxWidth (AutoWidth!)
BoxTextAngle (Angle:Degrees180!)
BoxHorizontalAlignment (Alignment:AlignMargins!; Position:Right!)
BoxVerticalAlignment (Alignment:Bottom!)
BoxTextEdit()
Type (ReturnAdd)
Close ()
Type (" ")
DeleteCharPrevious ()

// Create TextBox for Send To address
BoxCreate (TextBox!)
BoxBorder (Style: 0)
BoxAttachTo (Anchor:Page!)
BoxWidth (AutoWidth!)
BoxTextAngle (Angle:Degrees180!)
BoxHorizontalAlignment (Alignment:AlignMargins!; Position:Center!;Offset:-.75")
BoxVerticalAlignment (Alignment:Center!)
BoxTextEdit ()
Type (SendToAdd)
Close ()
Type (" ")
DeleteCharPrevious ()
Quit

Label (CancelLabel)
Close (Save:No!)
Quit



Label (AddBook)
if (AddBook[5] = 274)

CancelLabel
endif
Switch (AddBook[3])
CaseOf "PushBttn10":

AddressBookDlg ()
ReturnAdd = AddressBookSelectionFormatAddr (AddressBookFormatGetCurrent (Item:Format!))
RegionSetWindowText ( NamedRgn:"Address.Edit3"; Item:ReturnAdd)
Return

CaseOf "PushBttn11":
AddressBookDlg ()
SendToAdd = AddressBookSelectionFormatAddr (AddressBookFormatGetCurrent (Item:Format!))
RegionSetWindowText ( NamedRgn:"Address.SendToAdd"; Item:SendToAdd)
Return

CaseOf "CancelBttn":
DialogDismiss (Dialog:"Address"; Control:"CancelBttn")
CancelLabel

CaseOf "OKBttn":
DialogDismiss (Dialog:"Address"; Control:"OKBttn")
CallbackResume

EndSwitch


